
CHAPTER 4. AN INTRODUCTION TO THE WRITING AND 

SIMULATION OF GENERAL EQUILIBRIUM MODEL 
 

1. A very simple general equilibrium model 

 

The model has one household and one firm.  

The household has a fixed stock of labor (L), which is entirely supplied to the labour market if 

the wage rate (W) is positive.  

There is a single good (X) produced from the single input labor. The firm buys labor services 

and sells X to the household at price P.  

The household receives income from selling labor services and uses it all to buy X. 

 

In physical units, the household supplies labor to the firm and receives X output. In payments, 

the household receives labor income from the firm and pays the firm for X. 

 

The consumer has no alternative use for labor and so optimizes by supplying it all. The 

consumer prefers more X and so the demand for X is going to be given by X = WL/P: it is 

optimal to spend all income on X. 

 

The production function for X is given as LX α= , where α  is the marginal product of labor 

in X production. We assume competition and free entry, so that the firm is forced to price at 

marginal cost. Marginal cost, the cost of producing one more unit of X, is given by α/W  

( α/1  is the amount of labor needed for one unit of X). 

 

Our model has only two parameters, total labor supply (LBAR) and productivity α  (ALPHA). 

 

PARAMETERS 

LBAR labor supply (fixed and inelastic) 

ALPHA productivity parameter X = ALPHA*L; 

 

LBAR = 100; 

ALPHA = 2; 

There are four positive variables: price (P), quantity of X (X), the wage rate (W), and 

consumer income (INCOME). 

 

POSITIVE VARIABLES 

P price of X 

X quantity of X 

W wage rate 

INCOME income from labor supply; 

 

There are four equations.  

• The pricing equation of firms, which requires that marginal cost is greater than or 

equal to price, with output being the complementary variable to this inequality. With 

constant returns, marginal cost is also average cost, and so this equation can also be 

thought of as a free-entry, zero-profits condition. Accordingly, we refer to this 

equation as ZPROFIT (zero profits). 

• Then we require two market-clearing conditions. First, supply of X must be greater 

than or equal to demand with the complementary variable being P, the price of X. This 

is referred to as CMKTCLEAR (commodity-market clearing). Second, the supply of 

labor must be greater than or equal to its demand, with the complementary variable 

being W, the wage rate (price of labor). This is labeled LMKTCLEAR (labor-market 

clearing).  

• Finally, we require income balance: consumer expenditure equals labor income. This 

is labeled CONSINCOME. 

 

EQUATIONS 

ZPROFIT zeroprofits in X production 

CMKTCLEAR commodity (X) market clearing 

LMKTCLEAR labor market clearing 

CONSINCOME consumer income; 

 

The equations are very simple.  

• Marginal cost, equal average cost, is just the wage rate divided by α , the productivity 

parameter as discussed above.  

• Commodity market clearing requires that X in equilibrium is greater than or equal to 

demand, which is just income divided by P, the price of X.  

• Labor-market clearing requires that supply, LBAR, is greater than or equal to labor 

demand, given by X divided by " (X/").  

• Income spent on X must equal wage income, W times LBAR (W*LBAR). 

 

ZPROFIT.. W/ALPHA =G= P; 

CMKTCLEAR.. X =G= INCOME/P; 

LMKTCLEAR.. LBAR =G= X/ALPHA; 

CONSINCOME.. INCOME =G= W*LBAR; 

 

We name the model “GE” for general equilibrium, and associate each equation in the model 

with its complementary variable. 

 

MODEL GE /ZPROFIT.X, CMKTCLEAR.P, LMKTCLEAR.W, 

CONSINCOME.INCOME/; 

 

Now we introduce a new feature, namely, setting starting values of variables. This helps the 

solver find the solution, and can be quite important in complex problems. The notation for 

setting an initial value of a variable is the NAME.L notation we used earlier. 

 

* set some starting values 

P.L = 1; 

W.L = 1; 

X.L = 200; 

INCOME.L = 100; 

 

One final issue that is familiar to students of economics, is that there is an indeterminacy of 

the “price level” in this type of problem. If P = W = 1 and INCOME = 100 are solutions to 

this model, so are any proportional multiples of these values such as P = W = 2, INCOME = 

200. More formally, there are really only three independent equations in this model, the fourth 

is automatically satisfied if three hold. This is known in economics as “Walras’ Law”. 

The solution to this problem is to fix one price, termed the “numeraire”. Then all prices are 

measured relative or in terms of this numeraire. Suppose that we choose the wage rate to be 

the numeraire. The notation is “W.FX” where .FX stands for “fix”. It is important to 

understand that W.FX is not the same as W.L. The former holds W fixed throughout the 

remainder of the program, whereas W.L is just setting an initial value that will be changed by 

the solver (unless it happens to be the correct equilibrium value of course). When a variable is 

fixed, GAMS automatically drops the complementary equation from the model. 

 

* choose a numeraire 

W.FX = 1; 

 

Now solve the model. As a counter-factual experiment, double labor productivity and resolve. 

 

SOLVE GE USING MCP; 

 

* double labor productivity 

ALPHA = 4; 

SOLVE GE USING MCP; 

 

Here is the solution to the first solve statement. Notice that W.FX shows up as fixing both the 

upper and lower bounds for W (line 196) 

 

                       LOWER     LEVEL     UPPER    MARGINAL 

 

---- VAR P               .        0.500     +INF       .          

---- VAR X               .      200.000     +INF       .          

---- VAR W              1.000     1.000     1.000      EPS        

---- VAR INCOME          .      100.000     +INF       .     

 

Here is the solution to the second solve statement. Doubling of labor productivity allows 

twice as much X to be produced from the fixed supply of labor. With the wage rate fixed at W 

= 1, the equilibrium price of X falls in half: twice as much X can be purchased from the  

income from a unit of labor (line 359). 

                       LOWER     LEVEL     UPPER    MARGINAL 

 

---- VAR P               .           0.250        +INF          .          

---- VAR X               .      400.000         +INF         .          

---- VAR W              1.000     1.000      1.000        EPS        

---- VAR INCOME    .        100.000     +INF       .          

 

2. A general formulation of general equilibrium models 

 

First, general-equilibrium models consist of activities that transform some goods and factors 

into others. These include outputs of goods from inputs, trade activities that transform 

domestic into foreign goods and vice versa, activities that transform leisure into labor supply, 

and activities that transform goods into utility (welfare). 

 

Activities are most usefully represented by their dual, or cost-functions. The conditions for 

equilibrium are then that marginal cost for each activity is greater than or equal to price, with 

the complementary variable being the equilibrium quantity or “level” that activity. A quantity 

variable is complementary to a price equation. With competition and constant returns to scale, 

these conditions are also referred to as zero profit conditions. 

 

Second, general-equilibrium models consist of market clearing conditions. A commodity is a 

general term that includes goods, factor of production, and even utility. Thus X and labor are 

both commodities in our example. Activities transform some commodities into other 

commodities. Market clearing conditions require that the supply of a commodity is greater 

than or equal to its demand in equilibrium, with the complementary variable being the price of 

that commodity. A price variable is complementary with a quantity equation. 

 

Finally, there are income-balance equations for each “agent” in a model. Agents are generally 

household, but often include a government sector, or the owner of a firm in models with 

imperfect competition and pure profits. Expenditure (Exp) equals income for each agent. 

 

Let i subscript activities, also referred to as production sectors. Let j subscript commodities, 

and let k subscript agents or households. Our general formulation is then: 

The general-equilibrium system then consists of i+j+k inequalities in i+j+k unknowns: i 

activity levels, prices of j commodities, and the incomes/expenditures of k agents. As per our 

earlier discussion, the price level is indeterminate and one price is fixed as a numeraire. 

 

 

 

 

 

 

 

 

 

 

 

 

 

 


